Newsgroup: it.comp.appl.macromedia
Titolo: D7 problema script per sfogliare immagini e password per "porta segreta"

Salve a tutti ed un grazie a coloro che mi daranno una mano.
Sto realizzando per la prima volta un cd multimediale. Il cd non sarà
a scopo di lucro ma per un'attività ricreativa. Uno studio mi ha messo
a disposizione il Director 7 ma non mi hanno anche intimato
"Arrangiati !". Allora mi sono letto un manuale sugli usi generici
degli script con il Lingo e altre generalità del D7 e mi sono fatto un
quadro della situazione che mi ha permesso di arrivare fino a qui
arrangiandomi da solo.  Mi sono però arenato miseramente davanti a due
problemi :
	1) vorrei avere una pagina in cui, utilizzando due pulsanti per
avanzare o indietreggiare, si potesse sfogliare delle immagini come se
fosse un catalogo. Pensavo si potesse fare utilizzando un cast esterno
con i soli file di immagine e assegnare al comando sprite una
variabile globale in modo da scorrere all'interno del cast ma la cosa
non funziona... credo di non aver capito bene come programmare i due
pulsanti...
	2) volevo anche dare la possibilità di accedere ad una pagina segreta
semplicemente inserendo una password da tastiera senza però vedere le
lettere che si inseriscono. C'è già un beaviour pronto che pensavo con
qualche modifica si potesse arrangiare ma la cosa di nuovo non
funziona...
	Attualmente non ho le conoscenze per aggirare il primo ostacolo mentre
per il secondo potrei eventualmente creare un' "hot spot" su una
qualsiasi schermata che sarebbe facilmente individuabile.

 wrote in message 8gg5c4$3uh$1...@serv1.iunet.it...
Salve a tutti ed un grazie a coloro che mi daranno una mano.
Sto realizzando per la prima volta un cd multimediale. Il cd non sarà
a scopo di lucro ma per un'attività ricreativa. Uno studio mi ha messo
a disposizione il Director 7 ma non mi hanno anche intimato
"Arrangiati !". Allora mi sono letto un manuale sugli usi generici
degli script con il Lingo e altre generalità del D7 e mi sono fatto un
quadro della situazione che mi ha permesso di arrivare fino a qui
arrangiandomi da solo.  Mi sono però arenato miseramente davanti a due
problemi :
1) vorrei avere una pagina in cui, utilizzando due pulsanti per
avanzare o indietreggiare, si potesse sfogliare delle immagini come se
fosse un catalogo. Pensavo si potesse fare utilizzando un cast esterno
con i soli file di immagine e assegnare al comando sprite una
variabile globale in modo da scorrere all'interno del cast ma la cosa
non funziona... credo di non aver capito bene come programmare i due
pulsanti...
....eccoti uno script fatto su misura da attaccare sui bottoni (mi devi una birra...)
------
property Fotografie, myCastFoto, numFoto, direzione, spriteFoto
global currentFotoNum
-- michele rota
-- The Brain box
-- (new media design)
-- 
www.thebrainbox.com
-- 
mic...@thebrainbox.com
on beginSprite me
  if voidP(currentFotoNum) then  currentFotoNum = 1
  fotografie = []
  numFoto = the number of members of castLib myCastFoto
  repeat with k = 1 to numFoto
    if the type of member k of castLib myCastFoto = #bitmap then
      add(Fotografie, the name of member k of castlib myCastFoto)
    end if
  end repeat
  numFoto = count(fotografie)
end
on endSprite me
  currentFotoNum = void
end
on mouseUp me
  if direzione = #avanti then  -- avanti
    currentFotoNum = currentFotoNum + 1
    if currentFotoNum <= numFoto then
      foto = getAt(Fotografie, currentFotoNum)
      set the member of sprite spriteFoto to member foto of castlib myCastFoto
    else
      currentFotoNum = numFoto
    end if
  else -- indietro
    currentFotoNum = currentFotoNum - 1
    if currentFotoNum >= 1 then
      foto = getAt(Fotografie, currentFotoNum)
      set the member of sprite spriteFoto to member foto of castlib myCastFoto
    else
      currentFotoNum = 1
    end if
  end if
end
-- attento che la pList va tutta su una linea...
on getPropertyDescriptionList me
  pList = [myCastFoto:[#comment:"Nome del cast con le foto:",#format:#String,#default:"nomedelcast"], direzione:[#comment:"dove vuoi
che scorra?", #format:#symbol, #range: [#avanti, #indietro], #default:#avanti], spriteFoto:[#comment:"Numero dello sprite Foto",
#format:#integer, #default:10]]
  return pList
end
-------